NASA TECHNICAL MEMORANDUM

NASA TM X-73528

NASA TM X-73528

(NASA-TM-X-73528) CENTROID AND MOMENTS OF AN AREA USING A DIGITIZER (NASA) 19 p HC A02/MF A01 CSCL 09B

N77-11718

Unclas

G3/61 54567

CENTROID AND MOMENTS OF AN AREA USING A DIGITIZER

by R. W. Patch
Lewis Research Center
Cleveland, Ohio 44135
October 1976

1. Report No.	2. Government Accession No.	3. Recipient's Catalog No.
NASA TM X-73528		

4. Title and Subtitle	5. Report Date
CENTROID AND MOMENTS OF AN AREA USING A DIGITIZER	
	6. Performing Organization Code
7. Author(s)	8. Performing Organization Report No.
R. W. Patch	E-8947
9. Performing Organization Name and Address	10. Work Unit No.
Lewis Research Center National Aeronautics and Space Administration Cleveland, Ohio 44135	11. Contract or Grant No.
12. Sponsoring Agency Name and Address	13. Type of Report and Period Covered
National Aeronautics and Space Administration Washington, D.C. 20546	Technical Memorandum
	14. Sponsoring Agency Code

15. Supplementary Notes

16. Abstract

The Centroid and Moments of an Area program provides the centroid, moments of inertia, product of inertia, radii of gyration, and area of any closed planar geometric figure. The figure must be available in graphic form and is digitized once with chart digitizer (graphic tablet). The digitizer origin may be set anywhere on the digitizer table. After digitizing, fifteen quantities are calculated and displayed: (1) area, (2) moment of inertia of area with respect to digitizer x-axis, (3) moment of inertia of area with respect to digitizer y-axis, (4) product of inertia of area with respect to digitizer axes, (5) first moment of x for digitizer axes, (6) first moment of y for digitizer axes, (7) x coordinate of centroid, (8) y coordinate of centroid, (9) moment of inertia of area with respect to x axis through centroid, (10) moment of inertia of area with respect to y axis through centroid, (11) product of inertia of area with respect to x and y axes through centroid, (12) polar moment of inertia of area around centroid, (13) radius of gyration about digitizer x axis, (14) radius of gyration about digitizer y axis, and (15) variance in the x-direction.

17. Key Words (Suggested by Author(s))	18. Distribution Statement
Digitizer; Centroid; Moment of inertia; Product of inertia; Radius of gyration; Variance; Graphic tablet; Desk calculator, Beam, Column, Mechanics of material	Unclassified - unlimited

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of Pages	22. Price*
Unclassified	Unclassified		

* For sale by the National Technical Information Service, Springfield, Virginia 22161

NASA-C-168 (Rev. 10-75)

CENTROID AND MOMENTS OF AN AREA USING A DIGITIZER

by R. W. Patch

Lewis Research Center

INTRODUCTION

The object of this program is to provide a general-purpose digitizer program for calculating area and first and second moments of any closed planar geometric figure such as a cross section of a beam or column or a spectral line shape that has been closed at the baseline. For simple figures such as circles, rectangles, or triangles, the area and moments are easily calculated by hand from simple formulas given in handbooks. For figures consisting of two or more simple figures superimposed, the area and moments may be calculated by hand with somewhat more difficulty. The principal application of this program, however, is to complicated or irregular figures such as spectral line shapes, beams, columns, rotating machine parts, statistical distributions, and ship hull sections, which are difficult or impossible to calculate by hand.

E-8947

In addition to the basic quantities of area and moments, certain derived quantities are also calculated for convenience.

Previous published digitizer programs have merely calculated area or circumference of closed geometric figures.

PROBLEM TASK DESCRIPTION

A closed planar geometric figure of arbitrary shape is shown graphically in figure 1. The size of the figure is such as to fit on the sensitive portion of the digitizer table (graphic tablet). The figure is to be digitized by going around it once clockwise with the cursor. Curved portions are to be digitized at closely spaced intervals (hereafter referred to as "continuous digitizing" because the intervals are determined automatically). To improve accuracy, straight portions may be digitized by merely digitizing the two end points.

The quantities to be calculated require several axes. The digitizer axes x and y may be set anywhere on the sensitive portion of the digitizer table. The closed figure has a centroid whose position is not known initially. Through the centroid are centroidal axes x_g and y_g parallel to the x and y axes, respectively. Consequently, if the angular orientation of the x_g and y_g axes relative to the figure is important, the figure should be rotated to the desired angular orientation relative to the digitizer table before doing the digitizing for this program. This rotation may be facilitated by using the cursor and another small program to check the figure angular orientation (ref. 1).

STAR category 09

The following quantities are to be calculated and displayed for each case.

(1) Area of figure

(2) First moment of x with respect to y axis

(3) First moment of y with respect to x axis

(4) Moment of inertia of area with respect to x axis

(5) Moment of inertia of area with respect to y axis

(6) Product of inertia of area with respect to x and y axes

(7) x coordinate of centroid

(8) y coordinate of centroid

(9) Moment of inertia of area with respect to x_g axis

(10) Moment of inertia of area with respect to y_g axis

(11) Product of inertia of area with respect to x_g and y_g axes

(12) Polar moment of inertia of area around centroid

(13) Radius of gyration about x axis

(14) Radius of gyration about y axis

(15) Variance in the x direction

Most of these quantities are defined in reference 2, but they are also defined in the next section or in the List of Symbols.

METHOD OF SOLUTION

Six basic integrals must be calculated.

$$A = \int dA \qquad (1)$$

$$c_x = \int x \, dA \qquad (2)$$

$$c_y = \int y \, dA \qquad (3)$$

$$I_x = \int y^2 \, dA \tag{4}$$

$$I_y = \int x^2 \, dA \tag{5}$$

$$I_{xy} = \int xy \, dA \tag{6}$$

(Symbols are defined in the List of Symbols.) From these integrals, all other required quantities may be found. To minimize storage requirements the six integrals are found by accumulating 12 sums as the closed figure is traversed with the cursor, assuming for curved portions that the figure consists of a succession of straight lines between digitized points. The six integrals are calculated from the 12 sums at the end of the program.

The area A may be expressed

$$A = \iint dy \, dx = \int_{\text{upper}} y \, dx - \int_{\text{lower}} y \, dx \tag{7}$$

where "upper" and "lower" refer to the parts of the figure for which a perpendicular outward vector is in the +y and -y directions, respectively. Using the trapezoidal rule, equation (7) becomes

$$A = \frac{1}{2} \sum (y_o + y_n)(x_n - x_o) \tag{8}$$

where subscripts o and n refer to the first (old) and second (new) points, respectively, of an adjacent pair of digitized points, and the summation is over all pairs of adjacent points all the way around the closed figure. For $x_n = x_o$ there is no contribution to equation (8).

The first moment of x with respect to the y axis is

$$c_x = \iint x \, dy \, dx = \int_{\text{upper}} xy \, dx - \int_{\text{lower}} xy \, dx \tag{9}$$

or

$$c_x = \frac{1}{2} \sum (y_o x_n - x_o y_n)(x_n + x_o) + \frac{1}{3} \sum (y_n - y_o)\left(x_n^2 + x_n x_o + x_o^2\right) \tag{10}$$

Again, there is no contribution if $x_n = x_o$.

The first moment of y with respect to the x axis is

$$c_y = \iint y \, dx \, dy = \int_{\text{right}} yx \, dy - \int_{\text{left}} yx \, dy \tag{11}$$

where "right" and "left" refer to the parts of the figure for which a perpendicular outward vector is in the $+x$ and $-x$ directions, respectively. Equation (11) becomes

$$c_y = \frac{1}{2} \sum (y_o x_n - x_o y_n)(y_n + y_o) - \frac{1}{3} \sum (x_n - x_o)\left(y_n^2 + y_n y_o + y_o^2\right) \tag{12}$$

There is no contribution to the last term when $x_n = x_o$.

The moment of inertia of area with respect to the x axis is

$$I_x = \iint y^2 \, dx \, dy = \int_{\text{right}} y^2 \, x \, dy - \int_{\text{left}} y^2 \, x \, dy \tag{13}$$

or

$$I_x = \frac{1}{3} \sum (y_o x_n - x_o y_n)\left(y_n^2 + y_n y_o + y_o^2\right)$$

$$- \frac{1}{4} \sum (x_n - x_o)(y_n + y_o)\left(y_n^2 + y_o^2\right) \tag{14}$$

There is no contribution to the last term when $x_n = x_o$.

The moment of inertia of area with respect to the y axis is

$$I_y = \iint x^2 \, dy \, dx = \int_{\text{upper}} x^2 \, y \, dx - \int_{\text{lower}} x^2 \, y \, dx \tag{15}$$

or

$$I_y = \frac{1}{3} \sum (y_o x_n - x_o y_n)\left(x_n^2 + x_n x_o + x_o^2\right)$$

$$+ \frac{1}{4} \sum (y_n - y_o)(x_n + x_o)\left(x_n^2 + x_o^2\right) \tag{16}$$

There is no contribution if $x_n = x_o$.

The product of inertia of area with respect to the x and y axes is

$$I_{xy} = \iint xy \, dy \, dx = \frac{1}{2}\left[\int_{\text{upper}} xy^2 \, dx - \int_{\text{lower}} xy^2 \, dx\right] \tag{17}$$

or

$$I_{xy} = \frac{1}{4}\sum \frac{(y_o x_n - x_o y_n)^2 (x_n + x_o)}{x_n - x_o}$$

$$+ \frac{1}{3}\sum \frac{(y_o x_n - x_o y_n)(y_n - y_o)\left(x_n^2 + x_n x_o + x_o^2\right)}{x_n - x_o}$$

$$+ \frac{1}{8}\sum \frac{(y_n - y_o)^2 (x_n + x_o)(x_n^2 + x_o^2)}{x_n - x_o} \tag{18}$$

There is no contribution if $x_n = x_o$ and, if this occurs, each term must be set equal to zero to avoid dividing by zero.

The other required quantities may be found from the six basic integrals. The x coordinate of the centroid is

$$\bar{x} = \frac{\int x \, dA}{A} = \frac{c_x}{A} \tag{19}$$

The y coordinate of the centroid is

$$\bar{y} = \frac{\int y \, dA}{A} = \frac{c_y}{A} \tag{20}$$

The moment of inertia of area with respect to the x_g axis is (ref. 2)

$$I_{gx} = I_x - A\bar{y}^2 \tag{21}$$

The moment of inertia of area with respect to the y_g axis is (ref. 2)

$$I_{gy} = I_y - A\bar{x}^2 \tag{22}$$

The product of inertia of area with respect to the x_g and y_g axes is (ref. 2)

$$I_{gxy} = I_{xy} - A\overline{xy} \tag{23}$$

The polar moment of inertia of area around the centroid is (ref. 2)

$$J_g = I_{gx} + I_{gy} \tag{24}$$

The radius of gyration about the x axis is (ref. 2)

$$k_x = \sqrt{\frac{I_x}{A}} \tag{25}$$

The radius of gyration about the y axis is (ref. 2)

$$k_y = \sqrt{\frac{I_y}{A}} \tag{26}$$

The variance in the x direction is (ref. 3)

$$s_x^2 = \frac{I_{gy}}{A} \tag{27}$$

PROGRAM DESCRIPTION

The program is written in BASIC for a Hewlett-Packard 9830A calculator and 9864A digitizer.

The flow chart for MOMENT is given in figure 2. First an initial point X1,Y1 is digitized. Then a loop is entered which is traversed once for each new digitized point. During each traverse the sums T1 and T2 are always added to, but the sums S0 through S9 are only added to if X2 is not equal to X3.

Closure occurs when the operator brings the cursor back sufficiently close to the initial point. To check if this has occurred, two tests are made each time the loop is traversed: (1) has any point been outside the closure distance E from the initial point? (this test prevents closure before we have moved appreciably away from the initial point at the start), (2) is the last point inside the closure distance E from the initial point? E is arbitrarily set at 0.03 inch, and the distances D from the initial point are approximated by

$$D = |x_n - x_i| + |y_n - y_i| \tag{28}$$

When both conditions are met, closure is forced by making the new point the same as the initial point, and a last pass is made by the calculator through the upper part of the program loop.

When the last pass is complete, the loop is exited, and the digitizer gives two beeps to signal closure. The six integrals are then calculated from the 12 sums, and the nine derived quantities are calculated. The six integrals and nine derived quantities are then displayed three at a time.

The listing of MOMENT is

```
1Ø   E = .Ø3
2Ø   F1 = Ø
3Ø   F2 = Ø
4Ø   ENTER (9,*) X1,Y1
5Ø   X2 = X1
6Ø   Y2 = Y1
7Ø   SØ = Ø
8Ø   S1 = Ø
9Ø   S2 = Ø
1ØØ  S3 = Ø
11Ø  S4 = Ø
12Ø  S5 = Ø
13Ø  S6 = Ø
14Ø  S7 = Ø
15Ø  S8 = Ø
16Ø  S9 = Ø
17Ø  T1 = Ø
18Ø  T2 = Ø
19Ø  ENTER (9,*) X3,Y3
2ØØ  B = Y2 + Y3
21Ø  G = Y2 * X3 - X2 * Y3
22Ø  Q = Y3 * Y3 + Y2 * Y2
23Ø  N = Q + Y2 * Y3
24Ø  T1 = T1 + B * G
25Ø  T2 = T2 + G * N
26Ø  IF X2 = X3 THEN 42Ø
27Ø  F = X3 - X2
28Ø  H = X3 + X2
29Ø  I = Y3 - Y2
3ØØ  M = X3 * X3 + X2 * X2
31Ø  L = M + X3 * X2
32Ø  SØ = SØ + B * F
33Ø  S1 = S1 + G * H
34Ø  S2 = S2 + I * L
35Ø  S3 = S3 + G * L
36Ø  S4 = S4 + I * H * M
37Ø  S5 = S5 + F * N
38Ø  S6 = S6 + F * B * Q
39Ø  S7 = S7 + G * G * H/F
4ØØ  S8 = S8 + G * I * L/F
41Ø  S9 = S9 + I * I * H * M/F
42Ø  IF F2 = 1 THEN 55Ø
```

```
430 X2 = X3
440 Y2 = Y3
450 D = ABS(X3 - X1) + ABS(Y3 - Y1)
460 IF D > E THEN 480
470 GOTO 490
480 F1 = 1
490 IF D < E AND F1 = 1 THEN 510
500 GOTO 190
510 F2 = 1
520 X3 = X1
530 Y3 = Y1
540 GOTO 200
550 WRITE (9,*)
560 WAIT 300
570 WRITE (9,*)
580 A = S0/2
590 C1 = S1/2 + S2/3
600 I2 = S3/3 + S4/4
610 C2 = T1/2 - S5/3
620 I1 = T2/3 - S6/4
630 I3 = S7/4 + S8/3 + S9/8
640 G1 = C1/A
650 G2 = C2/A
660 I4 = I1 - A * G2 * G2
670 I5 = I2 - A * G1 * G1
```

```
680 I6 = I3 - A * G1 * G2
690 J = I4 + I5
700 K1 = SQR(I1/A)
710 K2 = SQR(I2/A)
720 V = I5/A
730 FIXED 2
740 DISP "XAVE"; G1; "VAR"; V; "A"; A
750 STOP
760 DISP "IX"; I1; "IY"; I2; "IXY"; I3
770 STOP
780 DISP "IGX"; I4; "IGY"; I5; "J"; J
790 STOP
800 DISP "IGXY"; I6; "CX"; C1; "CY"; C2
810 STOP
820 DISP "KX"; K1; "KY"; K2; "YAVE"; G2
830 END
```

The storage capacity required is 818 words.

OPERATING INSTRUCTIONS

Assuming the program is recorded on file 0 of a cassette, the following is a check list for operating MOMENT:

1. Turn on calculator and digitizer.
2. Insert cassette.
3. Line up axes of figure (if any) with axes of digitizer.
4. Press SCATCH A EXECUTE.
5. Press LOAD 0 EXECUTE.
6. Set origin.

7. Press RUN EXECUTE.

8. Place cursor at initial point of figure (anywhere on figure will do).

9. Press C on cursor to start continuous digitizing.

10. Move cursor around curve clockwise back to initial point.

11. Get two beeps to indicate closure.

12. Press C on cursor to turn off continuous digitizing.

13. Read $\overline{X}$, S^2, A.

14. Press CONT EXECUTE.

15. Read I_x, I_y, I_{xy}.

16. Press CONT EXECUTE.

17. Read I_{gx}, I_{gy}, J_g.

18. Press CONT EXECUTE.

19. Read I_{gxy}, c_x, c_y.

20. Press CONT EXECUTE.

21. Read k_x, k_y, $\overline{y}$.

22. Go back to step 6 or 7 for another case.

If the figure contains a straight segment, approach the segment in the continuous C mode, press C to turn off the digitizer at the beginning of the segment, go to the other end of the segment by any path desired, press C to turn on the digitizer at the end of the segment, and proceed around the rest of the figure. The corners of a polygon may be digitized one at a time by pressing S once at each corner.

A test case is given in figure 3, but should be laid out with drawing instruments. The correct results in appropriate powers of inches are

XAVE	1.00	VAR	0.67	A	8.00
IX	34.00	IY	13.33	IXY	14.67
IGX	7.11	IGY	5.33	J	12.44
IGXY	0.00	CX	8.00	CY	14.67
KX	2.06	KY	1.29	YAVE	1.83

However, do not expect to obtain accuracy to two decimal places on all quantities because the digitizer is only accurate to 0.01 inch and there is additional error in positioning the cursor.

APPENDIX - SYMBOLS

BASIC name	Display name	Mathematical symbol	Description
A	A	A	Area
B			$y_o + y_n$
C1	CX	c_x	First moment of x with respect to y axis
C2	CY	c_y	First moment of y with respect to x axis
D		D	Distance from initial point
E		E	Closure distance from initial point
F			$x_n - x_o$
F1			Flag to indicate if any point outside of closure distance
F2			Flag for last pass
G			$y_o x_n - x_o y_n$
G1	XAVE	$\overline{x}$	x coordinate of centroid
G2	YAVE	$\overline{y}$	y coordinate of centroid
H			$x_n + x_o$
I			$y_n - y_o$
I1	IX	I_x	Moment of inertia of area with respect to x axis
I2	IY	I_y	Moment of inertia of area with respect to y axis
I3	IXY	I_{xy}	Product of inertia of area with respect to x and y axes
I4	IGX	I_{gx}	Moment of inertia of area with respect to x_g axis
I5	IGY	I_{gy}	Moment of inertia of area with respect to y_g axis

BASIC name	Display name	Mathematical symbol	
I6	IGXY	I_{gxy}	Product of inertia of area with respect to x_g and y_g axes
J	J	J_g	Polar moment of inertia of area around the centroid
K1	KX	k_x	Radius of gyration about x axis
K2	KY	k_y	Radius of gyration about y axis
L			$x_n^2 + x_o^2 + x_n x_o$
M			$x_n^2 + x_o^2$
N			$y_n^2 + y_o^2 + y_o y_n$
Q			$y_n^2 + y_o^2$
S0			$\sum (y_o + y_n)(x_n - x_o)$
S1			$\sum (y_o x_n - x_o y_n)(x_n + x_o)$
S2			$\sum (y_n - y_o)\left(x_n^2 + x_n x_o + x_o^2\right)$
S3			$\sum (y_o x_n - x_o y_n)\left(x_n^2 + x_n x_o + x_o^2\right)$
S4			$\sum (y_n - y_o)(x_n + x_o)\left(x_n^2 + x_o^2\right)$
S5			$\sum (x_n - x_o)\left(y_n^2 + y_n y_o + y_o^2\right)$
S6			$\sum (x_n - x_o)(y_n + y_o)\left(y_n^2 + y_o^2\right)$
S7			$\sum \frac{(y_o x_n - x_o y_n)^2 (x_n + x_o)}{x_n - x_o}$

BASIC name	Display name	Mathe-matical symbol	
S8			$\sum \frac{(y_o x_n - x_o y_n)(y_n - y_o)\left(x_n^2 + x_n x_o + x_o^2\right)}{x_n - x_o}$
S9			$\sum \frac{(y_n - y_o)^2 (x_n + x_o)\left(x_n^2 + x_o^2\right)}{x_n - x_o}$
T1			$\sum (y_o x_n - x_o y_n)(y_n + y_o)$
T2			$\sum (y_o x_n - x_o y_n)\left(y_n^2 + y_n y_o + y_o^2\right)$
V	VAR	s_x^2	Variance in the x direction
X1,Y1		x_i, y_i	Coordinates of initial point
X2,Y2		x_o, y_o	Coordinates of old point
X3,Y3		x_n, y_n	Coordinates of new point
		x, y	Digitizer coordinates or axes
		x_g, y_g	Coordinates with respect to centroid or axes through centroid

REFERENCES

1. Fairman, Seibert; and Cutshall, Chester S.: Engineering Mechanics. Second ed. John Wiley & Sons, Inc., 1946.

2. Dixon, Wilfrid J.; and Massey, Frank J. Jr.: Introduction to Statistical Analysis. Second ed. McGraw-Hill Book Co., Inc., 1957.

3. Hewlett-Packard 9830A Calculator 11272B Extended I/O ROM, Operating Manual. Chapter 6, Hewlett-Packard, Calculator Products Division, 1972, pp. 6-3 to 6-4.



FIGURE 1. - CLOSED PLANAR GEOMETRIC FIGURE TO BE DIGITIZED, SHOWING DIGITIZER AXES AND CENTROIDAL AXES. THE FIGURE MAY BE ANY SHAPE.



FIGURE 2. - FLOWCHART FOR 'MOMENT'.



FIGURE 3. - TEST CASE (ALL DIMENSIONS IN INCHES).